```
> restart:
```

Ejercicio 2, examen parcial 08/04/2000

José Mª Goicolea, Abril 2000

Un hexaedro regular, de masa m y lado a , cuyos vértices son OPQRSTUV está colgado de su vértice O a un punto fijo A mediante un hilo inextensible y sin masa. El cubo está en reposo y una masa puntual m impacta en el punto B del mismo, que es el centro de la cara OSVR, con velocidad horizontal y paralela al plano OSUQ, de módulo v_0 . Tras el impacto la partícula queda completamente adherida al punto B.

Se pide:

- 1. Deducir el tensor central de inercia del cubo, referido (por ejemplo) a ejes paralelos a las aristas
- 2. Campo de velocidades del cubo en el instante inmediatamente posterior al impacto.
- 3. Percusión que se produce en el hilo como consecuencia del choque.

```
> with(linalg):
Warning, the protected names norm and trace have been redefined and unprotected
```

Masa del cubo

```
> masa:={m=rho*a^3};
```

$$masa := \{m = \rho\, a^3\}$$

Sean las direcciones (u, v, w) paralelas a las aristas. Momento de inercia de una rebanada perpendicular a w , respecto del eje Gw , normal a la misma:

```
> dI:=2*rho*int(int(v^2,v=-a/2..a/2),u=-a/2..a/2);
```

$$dI := \frac{1}{6}\rho\, a^4$$

Momento de inercia del cubo, integrando:

```
> Iww:=int(dI,w=-a/2..a/2);
```

$$Iww := \frac{1}{6}\rho\, a^5$$

En función de la masa total del cubo

```
> inercia:={lambda=subs(solve(masa,rho),Iww)};
```

$$inercia := \{\lambda = \frac{1}{6} m a^2\}$$

El eje x es horizontal, según la dirección de la velocidad inicial de la partícula v_0, el z vertical ascendente, y el y formando un triedro xyz a derechas. La expresión del vector velocidad de la partícula inicial es

```
> V0:=vector([v0,0,0]):
```

Las incógnitas del problema son los vectores velocidad de G y velocidad angular del cubo después del choque. Sabemos que la velocidad de G debe ser horizontal necesariamente, ya que el hilo impone esta misma condición a O, y el vector OG es vertical. Por tanto, establecemos la componente correspondiente igual a cero, eliminando una incógnita a priori:

```
> v[G]:=vector(3):v[G][3]:=0:evalm(v[G]);
```

```
> Omega:=vector(3):evalm(Omega);
```

$$[v_{G_1}, v_{G_2}, 0]$$

$$[\Omega_1, \Omega_2, \Omega_3]$$

Tensor de inercia esférico del cubo (en G):

```
> J[G]:=diag(lambda,lambda,lambda);
```

$$J_G := \begin{bmatrix} \lambda & 0 & 0 \\ 0 & \lambda & 0 \\ 0 & 0 & \lambda \end{bmatrix}$$

Vector GB

```
> r[GB]:=(a/2)*vector([-1/sqrt(6),-1/sqrt(2),1/sqrt(3)]);
```

$$r_{GB} := \frac{1}{2} a \left[-\frac{1}{6}\sqrt{6}, -\frac{1}{2}\sqrt{2}, \frac{1}{3}\sqrt{3} \right]$$

velocidad del punto B después de la impulsión (campo de velocidades del sólido):

```
> v[B]:=v[G]+crossprod(Omega,r[GB]);
```

$$v_B := v_G + \left[\frac{1}{6}\Omega_2\,a\sqrt{3} + \frac{1}{4}\Omega_3\,a\sqrt{2}, -\frac{1}{12}\Omega_3\,a\sqrt{6} - \frac{1}{6}\Omega_1\,a\sqrt{3}, -\frac{1}{4}\Omega_1\,a\sqrt{2} + \frac{1}{12}\Omega_2\,a\sqrt{6}\right]$$

Impulsión reactiva del hilo, necesariamente vertical (y positiva o nula; si el resultado final fuese negativo, el hilo se arrugaría, y habría que rehacer el cálculo poniendo $R = 0$).

\> **Rv:=vector([0,0,R]):**

Ecuación vectorial de balance de la cantidad de movimiento del conjunto cubo+partícula:

\> **eqv1:=evalm(m*V0+Rv)=evalm(m*v[B]+m*v[G]);**

$$eqv1 := [m\,v0, 0, R] =$$
$$\left[m\left(v_{G_1} + \frac{1}{6}\Omega_2\,a\sqrt{3} + \frac{1}{4}\Omega_3\,a\sqrt{2}\right) + m\,v_{G_1}, m\left(v_{G_2} - \frac{1}{12}\Omega_3\,a\sqrt{6} - \frac{1}{6}\Omega_1\,a\sqrt{3}\right) + m\,v_{G_2}, m\left(-\frac{1}{4}\Omega_1\,a\sqrt{2} + \frac{1}{1}\right.\right.$$

Las tres ecuaciones escalares a las que da lugar son:

\> **eq1x:=lhs(eqv1)[1]=rhs(eqv1)[1];**

\> **eq1y:=lhs(eqv1)[2]=rhs(eqv1)[2];**

\> **eq1z:=lhs(eqv1)[3]=rhs(eqv1)[3];**

$$eq1x := m\,v0 = m\left(v_{G_1} + \frac{1}{6}\Omega_2\,a\sqrt{3} + \frac{1}{4}\Omega_3\,a\sqrt{2}\right) + m\,v_{G_1}$$

$$eq1y := 0 = m\left(v_{G_2} - \frac{1}{12}\Omega_3\,a\sqrt{6} - \frac{1}{6}\Omega_1\,a\sqrt{3}\right) + m\,v_{G_2}$$

$$eq1z := R = m\left(-\frac{1}{4}\Omega_1\,a\sqrt{2} + \frac{1}{12}\Omega_2\,a\sqrt{6}\right)$$

Ecuación vectorial de balance del momento cinético en G del cubo, considerando el momento de la percusión $P = m\,v_G - R_v$

\> **eqv2:=evalm(crossprod(r[GB],m*v[G]-Rv))=evalm(J[G]&*Omega);**

$$eqv2 := \left[\frac{1}{4}a\sqrt{2}\,R - \frac{1}{6}a\sqrt{3}\,m\,v_{G_2}, \frac{1}{6}a\sqrt{3}\,m\,v_{G_1} - \frac{1}{12}a\sqrt{6}\,R, -\frac{1}{12}a\sqrt{6}\,m\,v_{G_2} + \frac{1}{4}a\sqrt{2}\,m\,v_{G_1}\right] = [\lambda\Omega_1, \lambda\Omega_2,$$

Ecuaciones escalares correspondientes

\> **eq2x:=lhs(eqv2)[1]=rhs(eqv2)[1];**

\> **eq2y:=lhs(eqv2)[2]=rhs(eqv2)[2];**

> **eq2z:=lhs(eqv2)[3]=rhs(eqv2)[3];**

$$eq2x := \frac{1}{4} a \sqrt{2}\, R - \frac{1}{6} a \sqrt{3}\, m\, v_{G_2} = \lambda \Omega_1$$

$$eq2y := \frac{1}{6} a \sqrt{3}\, m\, v_{G_1} - \frac{1}{12} a \sqrt{6}\, R = \lambda \Omega_2$$

$$eq2z := -\frac{1}{12} a \sqrt{6}\, m\, v_{G_2} + \frac{1}{4} a \sqrt{2}\, m\, v_{G_1} = \lambda \Omega_3$$

De estas últimas despejamos las componentes de Ω , y las sustituimos en las anteriores ecuaciones:

> **Omegas:=solve({eq2x,eq2y,eq2z},{Omega[1],Omega[2],Omega[3]});**

$$Omegas := \left\{ \Omega_3 = \frac{1}{12} \frac{a\, m\, (-\sqrt{6}\, v_{G_2} + 3\sqrt{2}\, v_{G_1})}{\lambda}, \Omega_2 = \frac{1}{12} \frac{a\, (2\sqrt{3}\, m\, v_{G_1} - \sqrt{6}\, R)}{\lambda}, \Omega_1 = -\frac{1}{12} \frac{a\, (-3\sqrt{2}\, R + 2 \ldots}{\lambda} \right.$$

> **eq1ax:=simplify(subs(Omegas,inercia,eq1x));**

> **eq1ay:=simplify(subs(Omegas,inercia,eq1y));**

> **eq1az:=simplify(subs(Omegas,inercia,eq1z));**

$$eq1ax := m\, v0 = \frac{13}{4} m\, v_{G_1} - \frac{1}{4}\sqrt{2}\, R - \frac{1}{4}\sqrt{3}\, m\, v_{G_2}$$

$$eq1ay := 0 = \frac{11}{4} m\, v_{G_2} - \frac{1}{4}\sqrt{3}\, m\, v_{G_1} - \frac{1}{4}\sqrt{3}\sqrt{2}\, R$$

$$eq1az := R = \frac{1}{4}\sqrt{2}\, (-2\sqrt{2}\, R + \sqrt{3}\, m\, v_{G_2} + m\, v_{G_1})$$

Estas tres ecuaciones permiten despejar los valores de (R, v_{G_1}, v_{G_2}) que resuelven el problema:

> **solu:=simplify(solve({eq1ax,eq1ay,eq1az},{R,v[G][1],v[G][2]}));**

$$solu := \{ v_{G_2} = \frac{5}{126}\sqrt{3}\, v0, v_{G_1} = \frac{41}{126} v0, R = \frac{1}{18}\sqrt{2}\, m\, v0 \}$$

Expresión de los vectores solución:

> **v[s][G]=simplify(subs(Omegas,inercia,solu,evalm(v[G])));**

$$v_{s_G} = \left[\frac{41}{126}\,v0, \frac{5}{126}\sqrt{3}\,v0, 0\right]$$

> **v[s][B]=simplify(subs(Omegas,inercia,solu,evalm(v[B])));**

$$v_{s_B} = \left[\frac{85}{126}\,v0, -\frac{5}{126}\sqrt{3}\,v0, \frac{1}{18}\sqrt{2}\,v0\right]$$

> **Omega[s]=simplify(subs(Omegas,inercia,solu,evalm(Omega)));**

$$\Omega_s = \left[\frac{1}{21}\frac{v0}{a}, \frac{17}{63}\frac{\sqrt{3}\,v0}{a}, \frac{3}{7}\frac{\sqrt{2}\,v0}{a}\right]$$

> **R[s]=subs(inercia,solu,evalm(Rv));**

$$R_s = \left[0, 0, \frac{1}{18}\sqrt{2}\,m\,v0\right]$$

> **P[s]=subs(inercia,solu,evalm(m*v[G]-Rv));**

$$P_s = \left[\frac{41}{126}\,m\,v0, \frac{5}{126}\sqrt{3}\,m\,v0, -\frac{1}{18}\sqrt{2}\,m\,v0\right]$$

Comprobación: balance cantidad de movimiento del conjunto

> **compr1:=subs(solu,evalm(m*V0+Rv))=subs(Omegas,inercia,solu,evalm(m*v[B]+m*v[G]));**

$$compr1 := \left[m\,v0, 0, \frac{1}{18}\sqrt{2}\,m\,v0\right] = \Bigg[$$

$$m\left(\frac{41}{126}\,v0 + \frac{\frac{1}{12}\left(\frac{41}{63}\sqrt{3}\,m\,v0 - \frac{1}{18}\sqrt{6}\sqrt{2}\,m\,v0\right)\sqrt{3}}{m} + \frac{1}{8}\left(-\frac{5}{126}\sqrt{6}\sqrt{3}\,v0 + \frac{41}{42}\sqrt{2}\,v0\right)\sqrt{2}\right) + \frac{41}{126}\,m\,v0,$$

$$m\left(\frac{2}{63}\sqrt{3}\,v0 - \frac{1}{24}\left(-\frac{5}{126}\sqrt{6}\sqrt{3}\,v0 + \frac{41}{42}\sqrt{2}\,v0\right)\sqrt{6}\right) + \frac{5}{126}\sqrt{3}\,m\,v0,$$

$$m\left(-\frac{1}{84}\sqrt{2}\,v0 + \frac{\frac{1}{24}\left(\frac{41}{63}\sqrt{3}\,m\,v0 - \frac{1}{18}\sqrt{6}\sqrt{2}\,m\,v0\right)\sqrt{6}}{m}\right)\Bigg]$$

> **simplify(compr1);**

$$\left[m\,v0, 0, \frac{1}{18}\sqrt{2}\,m\,v0\right] = \left[m\,v0, 0, \frac{1}{18}\sqrt{2}\,m\,v0\right]$$

Comprobación: balance momento cinético del conjunto

> **compr2:=crossprod(r[GB],m*V0)=subs(Omegas,inercia,solu,evalm(J[G] &*Omega+crossprod(r[GB],m*v[B])));**

$$compr2 := \left[0, \frac{1}{6}a\sqrt{3}\,m\,v0, \frac{1}{4}a\sqrt{2}\,m\,v0\right] = \left[\frac{1}{126}a\,m\,v0 - \frac{1}{4}a\sqrt{2}\,m\left(-\frac{1}{84}\sqrt{2}\,v0 + \frac{\frac{1}{24}\left(\frac{41}{63}\sqrt{3}\,m\,v0 - \frac{1}{18}\sqrt{6}\sqrt{2}\,m\right.}{m}\right.\right.$$

$$- \frac{1}{6}a\sqrt{3}\,m\left(\frac{2}{63}\sqrt{3}\,v0 - \frac{1}{24}\left(-\frac{5}{126}\sqrt{6}\sqrt{3}\,v0 + \frac{41}{42}\sqrt{2}\,v0\right)\sqrt{6}\right), \frac{1}{12}a\left(\frac{41}{63}\sqrt{3}\,m\,v0 - \frac{1}{18}\sqrt{6}\sqrt{2}\,m\,v0\right)$$

$$+ \frac{1}{6}a\sqrt{3}\,m\left(\frac{41}{126}v0 + \frac{\frac{1}{12}\left(\frac{41}{63}\sqrt{3}\,m\,v0 - \frac{1}{18}\sqrt{6}\sqrt{2}\,m\,v0\right)\sqrt{3}}{m} + \frac{1}{8}\left(-\frac{5}{126}\sqrt{6}\sqrt{3}\,v0 + \frac{41}{42}\sqrt{2}\,v0\right)\sqrt{2}\right)$$

$$+ \frac{1}{12}a\sqrt{6}\,m\left(-\frac{1}{84}\sqrt{2}\,v0 + \frac{\frac{1}{24}\left(\frac{41}{63}\sqrt{3}\,m\,v0 - \frac{1}{18}\sqrt{6}\sqrt{2}\,m\,v0\right)\sqrt{6}}{m}\right), \frac{1}{12}a\,m\left(-\frac{5}{126}\sqrt{6}\sqrt{3}\,v0 + \frac{41}{42}\sqrt{2}\,v0\right)$$

$$- \frac{1}{12}a\sqrt{6}\,m\left(\frac{2}{63}\sqrt{3}\,v0 - \frac{1}{24}\left(-\frac{5}{126}\sqrt{6}\sqrt{3}\,v0 + \frac{41}{42}\sqrt{2}\,v0\right)\sqrt{6}\right)$$

$$\left.+ \frac{1}{4}a\sqrt{2}\,m\left(\frac{41}{126}v0 + \frac{\frac{1}{12}\left(\frac{41}{63}\sqrt{3}\,m\,v0 - \frac{1}{18}\sqrt{6}\sqrt{2}\,m\,v0\right)\sqrt{3}}{m} + \frac{1}{8}\left(-\frac{5}{126}\sqrt{6}\sqrt{3}\,v0 + \frac{41}{42}\sqrt{2}\,v0\right)\sqrt{2}\right)\right]$$

> **simplify(compr2);**

$$\left[0, \frac{1}{6}a\sqrt{3}\,m\,v0, \frac{1}{4}a\sqrt{2}\,m\,v0\right] = \left[0, \frac{1}{6}a\sqrt{3}\,m\,v0, \frac{1}{4}a\sqrt{2}\,m\,v0\right]$$